Gillespie

4th April 2005

1 Results

IVA. Some Simple Systems

$$X \xrightarrow{c} Z \qquad (22)$$

The mean number of X molecules is given by:

$$X1(t) = X_0 e^{-ct}$$

Definition 1.1. Reaction (22)

Algorithm 1 Code for Reaction (22)

```
directive sample 10.0
directive plot X()

val c = 0.5
let X() = delay@c
run 1000 of X() (* 1000, 5000, 10000 *)
```



Figure 3. Simulation of reaction (22), with $c = 0.5$ and $X_0 = 1000$.



Figure 4. Simulation of reaction (22), with $c = 0.5$ and $X_0 = 5000$.



Figure 5. Simulation of reaction (22), with $c = 0.5$ and $X_0 = 10000$.

$$\overline{X} + Y \xrightarrow{c_1} 2Y \qquad (29a)$$
$$2Y \xrightarrow{c_2} Z \qquad (29b)$$

Definition 1.2. Reaction (29)

IVB. The Lotka Reactions

IVD. The Oregonator

Algorithm 2 Code for Reaction (29)

```
directive sample 5.0 10000
directive plot Y()

new c1@5.0:chan
new c2@0.000625:chan (* 0.0025, 0.000625 *)
let X() = ?c1; X()
let Y() =
  do !c1; Y(); Y()
  or !c2; ()
  or ?c2; ()
run (X() | 40 of Y()) (* (10,3000), (40,12000) *)
```



Figure 6. Simulation of reaction (29), with $c_1 = 5$ and $c_2 = 0.005$. Initial values $Y = 10$ and $Y = 3000$ respectively



Figure 7. Simulation of reaction (29), with $c_1 = 5$ and $c_2 = 0.00125$. Initial values $Y = 40$ and $Y = 12000$ respectively

$$\overline{X} + Y_1 \xrightarrow{c_1} 2Y_1 \qquad (38a)$$
$$Y_1 + Y_2 \xrightarrow{c_2} 2Y_2 \qquad (38b)$$
$$Y_2 \xrightarrow{c_3} Z \qquad (38c)$$

- Species Y_1 feeds on an inexhaustible food source $\overline{X}$ to reproduce (38a)
- Species Y_2 feeds on Y_1 to reproduce (38b)
- Species Y_2 can die of natural causes (38c)

Definition 1.3. Lotka Reaction (38)

Algorithm 3 Code for Lotka Reaction (38)

```
directive sample 30.0 10000
directive plot Y1(); Y2()

new c1@10.0:chan
new c2@0.01:chan
val c3 = 10.0

let X() = ?c1; X()
let Y1() =
  do !c1; Y1(); Y1()
  or ?c2
let Y2() =
  do !c2; Y2(); Y2()
  or delay@c3
run (X() | 1000 of Y1() | 1000 of Y2())
```



Figure 8. Simulation of Lotka reactions (38) with $c_1 = 10$, $c_2 = 0.01$ and $c_3 = 10$. Initial values $Y_1 = Y_2 = 1000$. Results for Y_1, Y_2 vs. t for $0 < t \leq 10$ and $0 < t \leq 30$, and for Y_2 vs. Y_1, respectively.



Figure 9. Repeated simulation of Lotka reactions (38) as in Figure 8.



Figure 10. Simulation of Lotka reactions (38) but with limited number of X species. Rates $c_1 = 0.0001$, $c_2 = 0.01$ and $c_3 = 10$. Initial values $X = 10^5$.



Figure 13. Simulation of Lotka reactions (38), but with limited number of X species and an additional reaction $Y_1 \xrightarrow{c_4} Z$ that allows the prey to die of natural causes. Rates $c_1 = 0.0002$, $c_2 = 0.01$ and $c_3 = 10$. Initial values $X = 10^5$.

$$\overline{X}_1 + Y_2 \xrightarrow{c_1} Y_1 \qquad (59a)$$

$$Y_1 + Y_2 \xrightarrow{c_2} Z_1 \qquad (59b)$$

$$\overline{X}_2 + Y_1 \xrightarrow{c_3} 2Y_1 + Y_3 \qquad (59c)$$

$$2Y_1 \xrightarrow{c_4} Z_2 \qquad (59d)$$

$$\overline{X}_3 + Y_3 \xrightarrow{c_5} Y_2 \qquad (59e)$$

Definition 1.4. Oregonator Reaction (59)

Algorithm 4 Code for Oregonator Reaction (59)

```
directive sample 6.0 10000
directive plot Y1(); Y2(); Y3()
new c1@2.0:chan
new c2@0.1:chan
new c3@104.0:chan
new c4@0.008:chan (* 0.016 / 2 *)
new c5@26.0:chan

let X1() = ?c1; X1()
let X2() = ?c3; X2()
let X3() = ?c5; X3()
let Y1() =
  do !c2
  or !c3; Y3(); Y1(); Y1()
  or !c4
  or ?c4
and Y2() =
  do !c1; Y1()
  or ?c2
and Y3() = !c5; Y2()
run (X1() | X2() | X3())
run (500 of Y1() | 1000 of Y2() | 2000 of Y3())
```



Figure 24. Simulation of Oregonator reactions (59) with $c_1 = 2$, $c_2 = 0.1$, $c_3 = 104$, $c_4 = 0.016$ and $c_5 = 26$. Initial values $Y_1 = 500$, $Y_2 = 1000$, $Y_3 = 2000$. (a) Y_1, Y_2, Y_3 vs. t for $0 \le t \le 6$. and (b) $1 \le t \le 3$. (d) Y_2 vs. Y_1 for $0 \le t \le 6$ (e) Y_3 vs. Y_1 (f) Y_3 vs. Y_2



Figure 26. Simulation of Oregonator reactions (59) as in Figure 24, but with a limited number of X_1 species. Rate $c_1 = 0.0002$. Initial values $X_1 = 10^4$. Results for Y_1, Y_2, Y_3 vs. t and X_1 vs. t, respectively.



Figure 27. Simulation of Oregonator reactions (59) as in Figure 24, but with a limited number of X_2 species. Rate $c_3 = 0.00104$. Initial values $X_2 = 10^5$. Results for Y_1, Y_2, Y_3 vs. t and X_2 vs. t, respectively.